


松山大学 経営学部

# 情報処理論（応用）



第20回 お絵かきプログラム



講師 檀 裕也

<http://www.cc.matsuyama-u.ac.jp/~dan/application/>

2006年12月 5日

# 出席確認

- 出席確認フォームから学籍番号および氏名を送信せよ。

## 出席確認フォーム


<http://www.cc.matsuyama-u.ac.jp/~dan/application/attendance.html>

# 今回の予定

- サンプルプログラム2
  - お絵かきプログラム
  
- 到達目標
  - マウスの状態を取得し、イベントハンドラを記述することによって、ラスタグラフィックスを描画するプログラムを作成できる。


# サンプルプログラム 2

- お絵かきプログラム
  - マウスを使って 絵を描くことができる
  - 既存画像を読み込んで、書き込むことができる
  - BMP形式、JPEGやPNGなどのファイル形式に対応




# 新しいプロジェクト

- プロジェクトの種類「Visual Basicプロジェクト」
- テンプレート「Windowsアプリケーション」
- プロジェクト名「ImageEditor」




# 画面の構成要素



- ツールボックスの中から必要なコントロールを選択し、フォームに貼り付ける。
  - コントロールをダブルクリック
  - コントロールをフォーム上にドラッグ & ドロップ
- MainMenu および PictureBox


# フォームの構成要素

ピクチャボックスを配置し、プロパティを設定する




PictureBox1

.Dock: Fill




# フォームの構成要素

メニューを設定し、イベントハンドラを記述する




## MainMenu

- MenuItem1: ファイル (&F)
- MenuItem2: 開く (&O)
- MenuItem3: 保存する (&S)
- MenuItem4: -
- MenuItem5: 終了 (&X)



## MainMenu

- MenuItem6: ペン (&P)
- MenuItem7: 太いペン
- MenuItem8: 標準のペン
- MenuItem9: 細いペン
- MenuItem10: -
- MenuItem11: 消去ペン





## MainMenu


- MenuItem12: 色 (&C)
- MenuItem13: パレット



# コンポーネントの追加




- 次のコンポーネントを配置する
  - OpenFileDialog
  - SaveFileDialog
  - ColorDialog




# Paint イベントの処理コード

- コード画面上部のプルダウンメニューから PictureBox および Paint イベントを選択する。



オブジェクトの選択



イベントの選択

# コードの記述 (1)

- グローバル変数の宣言
- フォームの Load イベントに対するコード
- PictureBox の Paint イベントに対するコード

```
Public Class Form1
    Inherits System.Windows.Forms.Form
    Dim bmp As Image ' Image クラスのオブジェクトとして bmp を宣言する
    Dim g As Graphics ' Graphics クラスのオブジェクトとして g を宣言する
    Dim p_width As Integer ' ペンの幅
    Dim p_color As Color ' ペンの色

    Windows フォーム デザイナで生成されたコード


    Private Sub Form1_Load(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MyBase.Load
        bmp = New Bitmap(PictureBox1.Width, PictureBox1.Height) ' PictureBox と同じ大きさのビットマップを生成
        g = Graphics.FromImage(bmp) ' g は bmp に対する Graphics オブジェクト
        g.Clear(PictureBox1.BackColor) ' 背景色で塗りつぶす
        MenuItem7.Checked = False ' 太いペン
        MenuItem8.Checked = True ' 標準のペン
        MenuItem9.Checked = False ' 細いペン
        p_width = 4
        p_color = Color.Black
    End Sub

    Private Sub PictureBox1_Paint(ByVal sender As Object, ByVal e As System.Windows.Forms.PaintEventArgs) Handles PictureBox1.Paint
        If Not (bmp Is Nothing) Then
            e.Graphics.DrawImage(bmp, 0, 0, bmp.Width, bmp.Height)
        End If
    End Sub
End Class
```

# コードの記述 (2)

- [開く]コマンドの処理
  - 「ファイルを開く」で選択した画像ファイルを表示する

```
Private Sub MenuItem2_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MenuItem2.Click
    'フィルタ条件を決める
    OpenFileDialog1.Filter = "ビットマップ形式 (*.bmp)|*.bmp"
    OpenFileDialog1.Filter &= "|すべてのファイル (*.*)|*.*"
    'ダイアログで選択した画像ファイルを開く
    If OpenFileDialog1.ShowDialog() = DialogResult.OK Then
        bmp = New Bitmap(OpenFileDialog1.FileName)
        g = Graphics.FromImage(bmp)
        PictureBox1.Image = bmp
    End If
End Sub
```



# コードの記述 ( 3 )

- [保存する] コマンドの処理
  - 画像ファイルの形式に応じて、Image オブジェクトに対する Save メソッドの適用を分ける。

```
Private Sub MenuItem3_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MenuItem3.Click
    'フィルタ条件を決める
    SaveFileDialog1.Filter = "ビットマップ形式 (*.bmp)|*.bmp"
    SaveFileDialog1.Filter &= "|GIF形式 (*.gif)|*.gif"
    SaveFileDialog1.Filter &= "|JPEG形式 (*.jpg)|*.jpg"
    SaveFileDialog1.Filter &= "|PNG形式 (*.png)|*.png"

    '拡張子に応じて保存形式を変える
    If SaveFileDialog1.ShowDialog() = DialogResult.OK Then
        If SaveFileDialog1.FileName.EndsWith(".bmp") Then
            bmp.Save(SaveFileDialog1.FileName, Imaging.ImageFormat.Bmp)
        ElseIf SaveFileDialog1.FileName.EndsWith(".gif") Then
            bmp.Save(SaveFileDialog1.FileName, Imaging.ImageFormat.Gif)
        ElseIf SaveFileDialog1.FileName.EndsWith(".jpg") Then
            bmp.Save(SaveFileDialog1.FileName, Imaging.ImageFormat.Jpeg)
        ElseIf SaveFileDialog1.FileName.EndsWith(".png") Then
            bmp.Save(SaveFileDialog1.FileName, Imaging.ImageFormat.Png)
        End If
    End If
End Sub
```

# 解説

- 文字列の末尾を判定する EndsWith メソッド

```
Dim s As String = "明日は晴れるだろう。"
```

```
If s.EndsWith( "である。" ) Then
```

```
    文字列 s は "である。" で終わっている
```

```
Else
```

```
    文字列 s は "である。" で終わっていない
```

```
End If
```

- 「ファイル名の最後が ".bmp" である」という条件

```
SaveFileDialog1.FileName.EndsWith( ".bmp" )
```

# コードの記述 ( 4 )

- [終了] コマンドの処理

```
Private Sub MenuItem5_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MenuItem5.Click
    Me.Dispose()
End Sub
```

# コードの記述 ( 5 )

- [ペン] メニューにある各コマンドの処理
  - 選択されたペンにチェックを付ける
  - 選択されないペンからチェックを外す

```
' [太いペン] コマンド
```

```
Private Sub MenuItem7_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MenuItem7.Click  
    MenuItem7.Checked = True  
    MenuItem8.Checked = False  
    MenuItem9.Checked = False  
    p_width = 10  
End Sub
```

```
' [標準のペン] コマンド
```

```
Private Sub MenuItem8_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MenuItem8.Click  
    MenuItem7.Checked = False  
    MenuItem8.Checked = True  
    MenuItem9.Checked = False  
    p_width = 4  
End Sub
```

```
' [細いペン] コマンド
```

```
Private Sub MenuItem9_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MenuItem9.Click  
    MenuItem7.Checked = False  
    MenuItem8.Checked = False  
    MenuItem9.Checked = True  
    p_width = 2  
End Sub
```



# コードの記述 ( 6 )

- [消去ペン]コマンドの処理
  - チェックの有無を反転させる
  - チェックがあれば、背景色をペンに設定する
  - チェックがなければ、黒色をペンに設定する

```
' [消去ペン] コマンド
Private Sub MenuItem11_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MenuItem11.Click
    MenuItem11.Checked = Not MenuItem11.Checked
    If MenuItem11.Checked = True Then
        p_color = PictureBox1.BackColor
    Else
        p_color = Color.Black
    End If
End Sub
```

# コードの記述 (7)

- PictureBox1 に対する MouseDown イベント
  - マウスが押されたときの処理を記述する
  - マウスの位置が bmp オブジェクトの内部にあることを確認し、現在のペンの太さ・色で塗りつぶす。

'マウスを押したときの処理

```
Private Sub PictureBox1_MouseDown(ByVal sender As Object, ByVal e As System.Windows.Forms.MouseEventArgs) Handles PictureBox1.MouseDown
    Dim x, y As Integer
    Dim mybrush As Brush

    x = e.X
    y = e.Y

    If x < bmp.Width And y < bmp.Height Then
        mybrush = New SolidBrush(p_color)
        g.FillEllipse(mybrush, x - p_width \ 2, y - p_width \ 2, p_width, p_width)
        PictureBox1.Refresh()
    End If
End Sub
```

# コードの記述 ( 8 )

- PictureBox1 に対する MouseMove イベント
  - 「マウスの左ボタンが押されている」という条件  
`e.Button = MouseButton.Left`
  - マウスをドラッグしていれば、すでに記述した PictureBox1\_MouseDown イベントハンドラを呼び出して描画処理をする。

```
'マウスをドラッグしたときの処理
Private Sub PictureBox1_MouseMove(ByVal sender As Object, ByVal e As System.Windows.Forms.MouseEventArgs) Handles PictureBox1.MouseMove
    If e.Button = MouseButton.Left Then
        PictureBox1_MouseDown(sender, e)
    End If
End Sub
```


# コードの記述 ( 9 )

- [パレット] コマンドの処理
  - ペンの色を「色の設定」ダイアログで選択する



```
' [パレット] コマンド
Private Sub MenuItem13_Click(ByVal sender As System.Object, ByVal e As System.EventArgs) Handles MenuItem13.Click
    ' ダイアログで選択した色を選択する
    If ColorDialog1.ShowDialog() = DialogResult.OK Then
        p_color = ColorDialog1.Color
    End If
End Sub
End Class
```



# 実行 (デバッグ)



- [デバッグ] [開始]でプログラムを実行する
- ツールバーの**開始ボタン**をクリックしてもよい



# まとめ

- マウスの状態を取得し、イベントハンドラを記述することによって、ラスタグラフィックスを描画するプログラムを作成した。

# 次回予定

- ソフトウェア開発と仕様書
  - 要求定義とソフトウェアの仕様
  - ソフトウェア設計書の作成
  - ソフトウェアの配布
- 自主課題として制作するプログラムの内容を考えておいてください。
- 12月 8日(金)は、出張のため休講します。